
US Training Days: Thursday 10" and Friday 11" November Backfo top
These training days will take place at the Holiday Inn Rosslyn at Key Bridge, 1900 N Fort :

Myer Drive, ARLINGTON, VA just prior to the Society for Neuroscience meeting in 3
Washington, DC. The days are suitable for both existing and prospective users so join us

and learn how to make the most of Spike2 and Signal for your individual data acquisition

and analysis requirements.

News

e Covers all the major features of Spike2 and Signal

¢ In-depth tutorials take you from recording data through to advanced
analysis techniques

o Optional workshop sessions allow you to discuss specific questions with
CED staff in a small group environment

Attendance is limited so please reqister early to reserve your place on the appropriate
course.

Future meetings and events

Spanish Neuroscience Society
University of Salamanca,

Salamanca, SPain
September 28" — 30" 2011

Society for Neuroscience 2011

Booth 2238 - Walter E. Washington Convention Center,
Washington, DC, USA

November 12" — 16" 2011

Australian Neuroscience Society 2012
Jupiters Hotel,

Gold Coast, QLD, Australia

January 29" — February 1% 2012

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are
also available.

Spike2 downloads | Released | Signal downloads Released
Spike2 version 7.07 07/11 Signal version 5.03 09/11
Spike2 version 6.16 03/11 Signal version 4.09 08/11
Spike2 demo Signal demo



http://www.holidayinn.com/hotels/us/en/arlington/waswp/hoteldetail
http://www.holidayinn.com/hotels/us/en/arlington/waswp/hoteldetail
http://www.ced.co.uk/nw9y.htm
http://www.senc.es/
http://www.sfn.org/am2011/
http://www.sallyjayconferences.com.au/ans2012/index.html
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu7u.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw5u.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigdemu.shtml

Scripts Spotlight

Gated analysis

One of the first things that you learn as a new Spike2 user is how to take measurements
over a time range by bracketing it with cursors and using the Cursor Regions dialog.
This is all well and good but can become tedious if you have multiple time ranges to
analyse. There are several scripts in the Spike2 scripts folder that can speed things up.

TimeRanges.s2s allows you to mark multiple time ranges of interest as coloured bars
using the state channel draw mode. You can then use WMeasure.s2s to take your
chosen cursor regions type measurement (sum, mean, peak-to-peak etc.) repeatedly at
fixed intervals over a selected time range or only during the marked time ranges, i.e.,
gated by the range marker channel. The results are tabulated in the log view and can be
pasted to a spreadsheet. You also have the option to plot the results as an additional
channel in the data file.

Similarly, GatedPower. s2s allows you to generate a power spectrum based only on the
data within marked time ranges. This script also has an option to overlap FFT blocks so
that all of the raw data is included in the spectrum. This option is given because power
spectra are performed on successive time slices or epochs with a duration determined by
the sample rate and the chosen FFT size (i.e. 256, 512, 1024 etc... samples). Partial
epochs cannot be analysed. So, if the duration of a range is not an exact multiple of the
epoch duration, then we have to make a choice between ignoring the data in the partial
epoch at the end of the time range or, overlapping the epochs for a perfect fit. In practice
we aim not for minimum overlap but an overlap of close to fifty percent.

Each of these scripts generates a hotkey on the Script Bar when you run them for the
first time. These buttons allow you to quickly flip from one script to the other. You will find
user guides in the comments section at the top of the script file. Users of Spike2 v7 will
find these scripts in the scripts folder.

If you have any questions please contact geoff@ced.co.uk

Q. I'd like to set-up my data-file so that different channels are displayed in different
colours for easy identification.

A. You can set individual colours for almost all types of data, channel and view in
Spike2. Click on the colour palette icon in the main Spike2 toolbar or select
Change colours.. from the Vview menu to access the Set colours dialog.

Set colours dialog

With a data file open, the Set colours dialog will display a list of the channels
in the file. The drop-down menu can then be used to select the different channel
elements such as primary, secondary and background colour for the listed

Back to top

CED
Default colours used in Spike2 can be changed by selecting Application colours in the drop-down menu of the Set colours dialog


channels. Click on the name of the channel you want to change and then click a
colour in the palette to apply it to the selected channel element.

Set custom colours

The available colours in the palette can be changed by double clicking on a
colour. This opens a new dialog where you can select Basic colours to add to the
palette or define your own Custom colours with user-definable settings for Hue
Saturation, Luminescence or RGB ratio.

Channel colour settings are stored in the resource file that is associated with the
data file. Resource files have the same name as the data file and are stored in
the same location with the extension .S2R. You can apply colour settings from
one file to other data files using the File menu Resource Files > Apply
Resource Files.. option.

Scripts: Spike2

Q.

| would Iik@expod a channel of waveform data to Matlab for external analysis.
The problem I'm having is that my original recording took place over a long
period, making the data file, and associated Matlab file, rather large. Is there any
way to export just sections of the waveform, perhaps based on a stimulus trigger
recorded as an event channel?

The script, TrigbataExport.s2s, will automate the export of sections of a
waveform channel, based around a trigger channel, to individual Matlab files. The
script allows the user to set the trigger channel and source waveform along with
a time range of data to export. The exported sections of data are saved to the
same folder location with the same filename as the original file and a suffix
indicating the count of the sections exported. The files also include a TextMark
channel with a marker holding the trigger time from the original file.

Back to top

Right-click the script icon
and save to disk.

If'you have any problems
opening the embedded
scripts in this newsletter
please let us know.



'$TrigDataExport|Script to export triggered sections of data to individual matlab files  



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 12/09/11



var drive$,path$,name$;

var num$;

var dataFile%;

var curr:=0.0;

var start;

var code%[4];

var noSteps%;

var c%;

var toneTime$;

var output%;

var markCh%;

var wave%;

var data[10000];

var binSz;

var size%;

var range;

var offSet;

var ok%;;

var waveCh%,trigCh%;



if ViewKind() = 0 then              'Check we have a data file open

dataFile%:= View();

FrontView(dataFile%);

else

dataFile%:= FileOpen("",0);        'If not prompt user to select one

WindowVisible(1);

endif;



if dataFile% < 0 then

    Message("Could not open a data file!");     'Quit if no file opened

    halt;endif;



DlgCreate("Settings");

DlgChan(1,"Waveform data",1);

DlgChan(2,"Trigger",46);

DlgReal(3,"Time range",0.0,10000);

DlgReal(4,"Offset",0.0,10000);

ok%:=DlgShow(waveCh%,trigCh%,range,offSet);

if ok%<1 then

    Message("Script cancelled");

    Halt;endif;

if range = 0 then

    Message("No time range selected!\nRun the script again and set a time range for export");

    Halt;endif;

binSz:= BinSize(waveCh%);

size%:= (1/binSz)*range;

resize data[size%];

drive$:= FileName$(1);

path$:= FileName$(2);

name$:= FileName$(3);



noSteps%:=Count(trigCh%,0.0,MaxTime());

for c%:=1 to noSteps% do

    FrontView(dataFile%);

    curr:= NextTime(trigCh%,curr,code%[]);

    toneTime$:=Print$("%f",curr);

    num$:=Print$("%d",c%);

    start:=curr-offSet;

    if start < 0 then

        start:=0;

        endif;

    ChanData(waveCh%,data[] ,start, start+range);

    

    output%:=FileNew(7,0,1,1,6,32);'New file holding channel data

    markCh%:= MemChan(8,32);

    wave%:= MemChan(1,0,binSz);

    

    MemSetItem(markCh%,0,0.0,code%[],toneTime$);

    ChanWriteWave(wave%,data[],0.0);

    

    MemSave(wave%,waveCh%);

    MemSave(markCh%,trigCh%);

    ChanDelete(wave%);

    ChanDelete(markCh%);

    ChanShow(waveCh%,trigCh%);

    ExportChanList(1); 'Initialise the channel list

    ExportChanList(0.0,MaxTime(),waveCh%,trigCh%);

    FileSaveAs(drive$+path$+name$+num$+".mat",100,0,"",0,

"UseSName=1|UseCName=1|WaveTimes=0|BinFreq=0|BinTimes=0|WaveData=1|MarkAs=1|TMarkAs=2|RMarkAs=2|RMarkData=1|WaveMarkAs=2|WaveMarkData=1|Compatability=0"); 'Export (string args)

    FileClose(0,-1);

    next;

Message("Export complete!");

CED
TrigDataExport.s2s


S

Q.

The script requires Spike2 version 6 or higher. It could also be modified to export
triggered data to other file formats if required.

ignal

S—

I'd be interested to know which models are supported by the dynamic clamp
system in Signal. In the past | have used a “home grown” system but the
programmer from our group has now left and so we no longer have any support
for our existing set-up.

The Dynamic Clamp system in Signal is configured from the Clamp tab of the
sampling configuration dialog. The clamp section of the sampling configuration
can be enabled or disabled from the Clamp tab of the Edit menu
Preferences dialog and provides the main setup dialog which is used to define
the models to use during an experiment. This dialog can be accessed on-line to
enable or disable models and to view, edit and update model parameters while
sampling. Up to 15 models of any type can be defined for use on-line in a single
sampling configuration. The number of models that are able to run
simultaneously during sampling is dependent on the sample rate and model
complexity.

Adding models to the Dynamic clamp setup

Below is a data file showing action potentials simulated using the combination of
models in the setup dialog above.

Simulated action potentials

CED
You can access the Dynamic clamp setup dialog on-line using the Models.. button in the clamping control toolbar. This toolbar also gives access to the Membrane analysis dialog and the holding current setting during clamp sampling


Hodgkin-Huxley

These models are used to simulate the membrane conductance generated by a
population of voltage-dependent ion channels that obey the standard Hodgkin-
Huxley equations:

o Hodgkin-Huxley (Alpha/Beta)- uses a formulation based upon multiple
rate equations

e Hodgkin-Huxley (Tau)- uses a formulation based on thresholds and
threshold sensitivities

Synapse
These are used to simulate synaptic current flow between two cells:

o Alpha synapse- uses the Alpha function to generate the post-synaptic
conductance profile

e CPG synapse- the pre-synaptic potential mediates the release of a
transmitter which causes a post-synaptic current to flow. This particular
model mimics the behaviour of synapses in the central pattern
generation region of the lobster central nervous system

o Destexhe synapse- uses the Destexhe model to generate the post-
synaptic conductance profile. The synapse can be activated by the pre-
synaptic potential being above or below a defined threshold

e Electrical synapse- simulates a simple electrical connection (a gap
junction) between two cells

o Exponential synapse- simulates the presence of a synapse using an
instantaneous rise time and a single exponential to mimic the post-trigger
decay of the conductance

o Exponential difference synapse- generates the post-synaptic
conductance profile based on the difference between two exponentials,
one used to model the rising phase and the other to model the decay

o User-defined synapse- generates a post-synaptic conductance based
on a user-defined waveform read from a text file

Leak models

These provide simpler dynamic clamp behaviours where the simulated
conductance is not time-dependent:

o Linear leak- represents current leakage through a cell membrane with
constant conductance

e Goldmann-Hodgkin-Katz (GHK) leak- simulates the behaviour caused
by diffusion of an ion through a membrane via always-open ion channels,
driven by the differing ionic concentrations either side of the membrane
and the voltage difference across it

e Boltzmann leak- resembles a linear leak with an additional factor in the
equation that represents a voltage-dependent blockage of the leak

Noise

This model can be used to study the effect of simulated noisy conductances on
neuronal behaviour:

e Ornstein-Uhelenbeck noise- generates a noise model using the mean
reverting stochastic process

Full details of the available models, and the mathematics used to generate them,
is available in the Signal on-line help.

Scripts: Signal m Back to top

Q. | have b sing the script, IntraSpikeAnalysis.sgs, which was included BFahd-click tR&SeTiptlicon
in eNewsletter 58, to take automated measurements from action potentials. This and save to disk.
works fine, but would it be possible to extend the script to give the half-height of T U e
each spike as well? opening the embedded

scripts in this newsletter
please let us know.


http://www.ced.co.uk/img/cednws58.pdf

'$IntraSpikeAnalysis v2|Script to automatically iterate through and measure multiple spike

'occurences in Signal data frames.  



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'Creates a virtual channel at a high sample rate and uses this for the analysis rather than the original

'data channel, thus avoiding any interpolation derived search errors. Consequently you need at least 

'Signal version 4 to run this script



'CED 15/09/11



var count%;					'Count of spikes in frame

var current;				'Current measured spike

var gotSpikeFlag% := 0;		'Flag if a spike is found

var frameCounter%;			'Keep count of frames

var data%;					

var chan% := 1;				'Channel to analyse

var spikeTimes[1000];		'Array to hold actual times of spikes

var printCount%;

var pTime,pAmp;				'Time and level of peak for spike					

var sTime,sAmp;				'Time and level of spike start	

var amp;					'Peak amplitude (pAmp-sAmp)

var RiseAmp10,RiseTime10;	'Level and time at 10% rise

var riseAmp90,RiseTime90;	'     "        "   90%  "				

var RiseAmp50,RiseTime50,FallTime50;	'Level and times for 50% rise and fall

var rTime;					'Time of steepest rising point

var hWidth;					'Half width time

var AHPTime,AHPAmp;			'Time and level of afterhyperpolarisation peak

var hHeight;

var maxdata,minData,pDAmp;	'range of data in initial frame. Peak detection amplitude set at range/20

var res%;					'Text file to hold detailed results

var test;					'Check to see if spikes present in frame

var check%;					'Flag to either inspect individual spikes or run through whole file				

var all%;					'Additional variable related to check%

var got%;					'Returned indicating search errors

var error%[7];				'Array of flags indication search errors

var numErrors%;				'Number of detected errors

var i%,err%;

var vChan%;

var ch$;

var pos$[7];				

var gFloat%[20];



pos$[0] := "Peak";			

pos$[1] := "Threshold";

pos$[2] := "10% rise";

pos$[3] := "90% rise";

pos$[4] := "50% rise";

pos$[5] := "50% fall";

pos$[6] := "AHP";

var ok%;					'Dialog return value

var rej%;					'Flag to ignore spike from analysis



HideAll();					'Hide script window and toolbars

ToolbarVisible(1);	



data% := FrontView();					'If current view is a data file use it

if Viewkind(data%) <> 0 then			'Otherwise select data file

	data% := FileOpen("",0,2);			'Prompt user to select file		

endif;



if data% < 0 then						'If unable to open a file

	Message("No file opened\nScript closing");		'Warn user

	Halt;								'End script

endif;				

					

Window(0,0,100,100);			'Maximise the data view

XRange(MinTime(),MaxTime());	'Show whole time range

FrontView(data%);				'Make sure data file is current

DlgCreate("Spike analysis");	

DlgChan(1,"Select channel to analyse",1);		'Select channel to analyse

DlgCheck(2,"Check individual spikes");			'Option to check individual spikes

ok% := DlgShow(chan%,check%);			

if ok% = 0 then					'If user Cancels

    Message("Cancel pressed\nScript closing");		'Warn user

	Halt;						'Halt script

endif;



Chanshow(chan%,1);



if App(-1) > 400 then;

    ch$:=Print$("%d",chan%);

    vChan%:=VirtualChan(0,"Ch("+ch$+")",0,0.00002,0);

else

    vChan%:=chan%;

endif;

    



res% := FileNew(1,0);			'Create a new text file for results

WindowTitle$("Results");		'..and title it

Window(0,70,100,100);			

Print("Frame\tSpike\tTime\tThreshold\tAmplitude\t10-90 Rise\tHalf width\tHalf height amp.\tAHP\n");	'Print headers to the result

View(data%);

CursorDelete(-1);

HCursorDelete(-1);		'Delete existing horizontal cursors

HCursorNew(chan%);				'Set horizontal cursor on channel 

Interact("Set horizontal cursor to cut upstroke of all spikes and press OK",1023); 'Allow user to set level

CursorSet(7);																		'Set cursors to show detected positions

CursorLabel(4,1,"Peak");															'Label cursors

CursorLabel(4,2,"Threshold");

CursorLabel(4,3,"10% rise");

CursorLabel(4,4,"90% rise");

CursorLabel(4,5,"50% rise");

CursorLabel(4,6,"50% fall");

CursorLabel(4,7,"AHP");

CursorLabelPos(1,10);																'Set label positions to avoid overlapping

CursorLabelPos(2,75);

CursorLabelPos(3,70);

CursorLabelPos(4,20);

CursorLabelPos(5,30);

CursorLabelPos(6,40);

CursorLabelPos(7,80);



MinMax(vChan%,MinTime(),MaxTime(),minData,maxData);			'Get full Y range of data in initial frame

pDAmp := (maxData-minData)/50.0;							'Calculate default value for use in peak detection



for frameCounter% := 1 to FrameCount() do			'Loop through each frame

	count% := 0;											'Reset spike counter

	Frame(frameCounter%);								'Go to frame	

	DrawAll();

	current := MinTime();								'Set variable to beginning of frame

	test := MinTime();												'Initial check to see if any spikes present

	if ChanSearch(vChan%,7,test,MaxTime(),HCursor(1)) = 0 then		'Print header based on presence of spikes in frame

		View(res%).Print("%d",frameCounter%);

	else

		View(res%).Print("%d\n\n",frameCounter%);

	endif;

	repeat													'Loop through each crossing	

		gotSpikeFlag% := ChanSearch(vChan%,7,current,MaxTime(),HCursor(1));

		if gotSpikeFlag% = 0 then							'If spike detected

			ArrConst(error%[],0);							'Reset error count

			err% := 0;										'Reset counter

			rej% := 0;										'Reset rejection flag

			pTime := current;								'Set peak time variable to start time

			got% := ChanSearch(vChan%,4,pTime,MaxTime(),0,pDAmp);	'Find spike peak

			if got% < 0 then error%[0] := 1;endif;			'Set value of error in peak find

			Cursor(1,pTime);								'Set cursor at peak position

			rTime := pTime;									'Set max slope time variable to peak time

			ChanSearch(chan%,9,rTime,rTime-0.005);			'Find maximum slope

			sTime := rTime;									'Set spike start variable to slope time

			got% := ChanSearch(vChan%,18,sTime,sTime-0.005,0,20);		'Find start of spike

			if got% < 0 then error%[1] := 1;endif;

			Cursor(2,sTime);								'Set cursor at spike start

			spikeTimes[count%] := sTime;					'Put time into array

			pAmp := ChanValue(vChan%,pTime);					'Get amplitudes at peak and start of spike

			sAmp := ChanValue(vChan%,sTime);

			amp := pAmp - sAmp;								'Get spike amplitude relative to start

			riseAmp10 := sAmp + (amp * 0.1);				'Calculate levels for 10%, 50% and 90% rise

			riseAmp90 := sAmp + (amp * 0.9);

			riseAmp50 := sAmp + (amp * 0.5);

			riseTime10 := sTime;									

			got% := ChanSearch(vChan%,7,riseTime10,pTime,riseAmp10);		'Find time at 10% rise

			if got% < 0 then error%[2] := 1;endif;

			Cursor(3,riseTime10);

			riseTime90 := sTime;

			got% := ChanSearch(vChan%,7,riseTime90,pTime,riseAmp90);		'Find time at 90% rise

			if got% < 0 then error%[3] := 1;endif;

			Cursor(4,riseTime90);

			riseTime50 := sTime;

			got% := ChanSearch(vChan%,7,riseTime50,pTime,riseAmp50);		'Find time at 50% rise

            if got% < 0 then error%[4] := 1;endif;

			Cursor(5,riseTime50);

            hHeight := ChanMeasure(vChan%,9,Cursor(2),Cursor(5));

			fallTime50 := pTime;

			got% := ChanSearch(vChan%,8,fallTime50,pTime + 0.005,riseAmp50);	'Find time at 50% fall

			if got% < 0 then error%[5] := 1;endif;

			Cursor(6,fallTime50);

			hWidth := fallTime50 - riseTime50;						'Half width time

			AHPTime := fallTime50;							

			got% := ChanSearch(vChan%,5,AHPTime,MaxTime(),0,pDAmp/5);		''Find trough following spike

			if got% < 0 then error%[6] := 1;endif;

			AHPAmp := ChanValue(vChan%,AHPTime);

			Cursor(7,AHPTime);

			count% += 1;									'Add 1 to spike count

			numErrors% := ArrSum(error%[]);

		 	if  numErrors% > 0 then

				check% := 1;

			endif;

			if check% > 0 then														

				XRange(current - 0.005,AHPTime + 0.005);

				if numErrors% > 0 then											'Routine to detect search errors and allow user to manually set positions

					all% := 1;													'Set variable to ensure we do not change display if spike rejected

					DlgCreate("SEARCH ERROR!!!",60,5);								'Set up dialog to be displayed if detection of any points failed

					DlgText("Failure in detection of following points",15,2);

					DlgText("May cause additional undetected errors",14,3);

					for i% := 1 to 7 do											'Display failed feature

						if error%[i%-1] > 0 then

							err% += 1;

							DlgText(pos$[i%-1],2,err%+4);

						endif;

					next;

					DlgText("Manually position cursors and continue, or reject spike from analysis",2,err%+6);

					DlgAllow(1023);

					DlgButton(1,"Continue");

					DlgButton(0,"Reject");

					ok% := DlgShow();

					if ok% = 0 then

						rej% := 1;

					endif;

				else

					all% := Interact("Check and adjust positions and proceed to next spike or run through all spikes",1023,0,"Next|0x0d","Run");

				endif;

				pAmp := ChanValue(vChan%,Cursor(1));					'Recalculate values based on current cursor positions

				sTime := Cursor(2);

				sAmp := ChanValue(vChan%,sTime);

				amp := pAmp - sAmp;

				riseTime10 := Cursor(3);

				riseTime90 := Cursor(4);

				riseTime50 := Cursor(5);

				fallTime50 := Cursor(6);

                hHeight := ChanMeasure(vChan%,9,sTime,riseTime50);

				AHPAmp := ChanValue(vChan%,Cursor(7));

				if all% = 2 then

					View(data%).XRange(MinTime(),MaxTime());

					DrawAll();

					check% := 0;

				endif;

			endif;

			if rej% = 0 then

				View(res%).Print("\t%d\t%g\t%g\t%g\t%g\t%g\t%g\t%g\n",count%,sTime*1000.0,sAmp,amp,(riseTime90-riseTime10)*1000.0,hWidth*1000.0,hHeight,AHPAmp);	'Print detailed results to text window

			else

				count% -= 1;

			endif;

		endif;													

	until gotSpikeFlag% < 0;							'Stop when no spike found		



	if count% > 0 then

		View(res%).Print("\n");

	endif;

next;

View(data%);

Window(0,0,100,70);

if App(-1) > 400 then;

FrontView(data%);

ChanDelete(vChan%);

endif;    

FrontView(res%);

RestoreAll();



proc HideAll()

var i%;

gFloat%[0] := App(-2);        'number of windows

for i% := 1 to gFloat%[0] do  'hide all windows and save state

   gFloat%[i%] := View(App(i%)).WindowVisible(0);

   next;

end



proc RestoreAll()

var i%;

for i% := 1 to gFloat%[0] do  'restore hidden windows

   View(App(i%)).WindowVisible(gFloat%[i%]);

   next;

end





CED
IntraSpikeAnalysis v2.sgs


The script, IntraSpikeAnalysis v2.sgs, has been extended from the
original version to include a measurement of the spike half-height alongside the
standard measurements, which are then printed to a text file. This script requires
Signal version 4 or higher.

Did you know...?

You can display channels in any order in a data or result file by clicking on the channel
number to the left of the Y-axis and dragging it to a new position.

Recent questions

Q.

I am currently using Signal, a Micro1401 and a 1902 along with a Magstim 2002
stimulator for experiments investigating cortical processing of speech sounds. I'm
recording EMG from a lip electrode but find that the TMS pulse introduces a large
artefact which makes measurement of the responses difficult. | am intending to
use the 1902 input clamp option to counter the artefact but would like to know if it
is possible to trigger the input clamp slightly before the TMS pulse occurs?

The input clamp option of the 1902 should always be activated before the
stimulator for optimal results. It can be triggered by either the rising or falling
edge of a TTL pulse delivered to the Trigger 2 input socket on the front panel.
You can set which edge of a pulse to use in the 1902 controls dialog. By
default the input clamp is set to use the rising edge of a pulse.

The Magstim2 range can also be set to trigger on either the rising or falling edge
of a pulse by wiring the trigger signal to the appropriate pin on the 26-way
connector on the rear of the unit (pin 5 = rising edge, pin 6= falling edge). In this
way you could use a single 1401 digital output wired to both Trigger 2 on the
1902 and pin 6 of the Magstim unit to trigger the input clamp on the rising edge of
the pulse and the stimulator on the falling edge, giving a suitable delay between
clamp and stimulus.

User group
You can now find our eNewsletters archived on the CED website.

Try the CED Forums bulletin board for software and hardware support

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

